
Arbeitszeitformular – Formeldokumentation
Feld H1:

=DATWERT(VERKETTEN("1.1.";$Basisdaten.B7))

Diese Formel sieht in den Basisdaten nach, welches Jahr dort eingetragen ist (Standard 
2006). Dann wird mit „Verketten“ das Datum des 1. Januar gebildet und schließlich der 
programminterne Datumswert errechnet. Über die Formatierung wird dann geregelt, dass 
in dem Feld der Monatsname in voller Länge angezeigt wird.

Feld I1:

=$Basisdaten.B7

in den Basisdaten wird das Jahr nach geschlagen

Feld K1:

=VERKETTEN($Basisdaten.A5;" ";$Basisdaten.B5)

Aus den Basisdaten werden das Wort „Personal-Nr.:“, ein Leerzeichen und die 
Personalnummer des Mitarbeiters verknüpft.

Feld N1:

=VERKETTEN($Basisdaten.A4;" ";$Basisdaten.B4)

Aus den Basisdaten werden das Wort „Name“, ein Leerzeichen und der Name des 
Mitarbeiters verknüpft.

Spalte A/B

In Zeile 6 wird zunächst die Zelle H1 abgefragt (erster Tag des Monats). In den folgenden 
Spalten wird dann jeweils ein Tag dazuaddiert. Über die Formatierung wird die Anzeige 
des Tages und des Wochentages geregelt.

Im Februar wird außerdem in der Zeile 34 geprüft, ob dieser Monat 29 Tage hat

=WENN(TAG(MONATSENDE($H$1;0))>28;B33+1;"")

Bei 28 Tagen erscheint in dieser Zeile keine Anzeige. Voraussetzung in Excel: Das 
Analyse-AddIn muss aktiviert sein



Spalte H

=WENN(ODER(D6="";E6="");"";WENN((((WERT(E6)-WERT(D6))*24)-
(R6+S6))>0;(((WERT(E6)-WERT(D6))*24)-(R6+S6));""))

Unter Verwendung der Hilfsspalten R (Pausen) und S (Tagesarbeitszeit) wird ermittelt, ob 
die in den Spalten D und E eingetragen Zeiten für Arbeitsbeginn und -Ende Mehr-
/Überstunden ergeben.

Spalte I

=WENN(ODER(C6="S";C6="N";H6="";$Basisdaten.$B$14="x";AJ6="";UND(
NICHT($Basisdaten.$B$12="");ODER(W6="";W6<$Basisdaten.$B$9)));"";W
ENN(ODER(C6="A";C6="T1";C6="F";C6="T2");AJ6*$Basisdaten.$B$29/100;
""))

Hier wird der Zuschlag berechnet:

Kein Zuschlag bei Spätschicht oder Nachtschicht. Ohne Mehrstunden (H) keine 
Zuschlagsberechnung. 

Bei Angestellten ohne Zuschlagsberechtigung wird kein Zuschlag berechnet 
(Basisdaten.B14 = x).  Ebenso keine Zuschlagberechnung bei Teilzeitkräften (Teilzeit-
Wochenarbeitszeit in Basisdaten.B12 eingetragen), solange die tarifvertragliche 
Wochenarbeitszeit nicht überschritten ist (Ergebnis von Hilfsspalte W entweder leer oder 
unter der tariflichen Wochenarbeitszeit).

Sonst wird ein Zuschlag in Höhe des Prozentwertes, der in den Basisdaten eingetragen 
ist, errechnet, sobald in Spalte C die Angaben „A“, „T1“, „F“ oder „T2“ auftauchen 
(Normalarbeitszeit, Tagschichten 1 und 2, Frühschicht). Der Zuschlag wird aber nur für die 
zuschlagpflichtigen Zeitanteile (Spalte AJ) errechnet

Spalte J

=WENN(ODER(NICHT(C6="S");H6="";$Basisdaten.$B$14="x";AJ6="";UND(
NICHT($Basisdaten.$B$12="");ODER(W6="";W6<$Basisdaten.$B$9)));"";W
ENN(C6="S";AJ6*$Basisdaten.$B$30/100;""))

Wie Spalte I nur mit Zuschlag für Spätschicht unter Ausschluss aller anderen Schichten

Spalte K

=WENN(ODER(NICHT(C6="N");H6="";$Basisdaten.$B$14="x";AJ6="";UND(
NICHT($Basisdaten.$B$12="");ODER(W6="";W6<$Basisdaten.$B$9)));"";W
ENN(C6="N";AJ6*$Basisdaten.$B$31/100;""))

Wie Spalte I nur mit Zuschlag für Nachtschicht unter Ausschluss aller anderen Schichten.



Feld L37

=WENN(H6="";"";SUMME(H6:K6))

Hier werden die Plusstunden aufsummiert (und in Zeile 37 die Gesamtsumme gebildet)

Feld I4

=$Basisdaten.$B29

Hier wird der Wert angezeigt, der für Überstunden tarifvertraglich gezahlt wird und in Zeit 
umgewandelt werden soll (Formatierung als Zahl mit Prozentzeichen). Dasselbe 
wiederholt sich in den Felder J4 und K4 (mit den Felder $Basisdaten.$B30 und 
$Basisdaten.$B31) 

Spalte N

=WENN(TEIL(C6;1;2)="ÜF";S6;WENN(ODER(D6="";E6="");"";WENN((((WE
RT(E6)-WERT(D6))*24)-(R6+S6))<0;((R6+S6)-((WERT(E6)-
WERT(D6))*24));"")))

Wenn in Spalte C die Buchstaben „ÜF“ (Überstundenfrei) enthalten sind, dann wird eine 
Tagesarbeitszeit (Spalte S) als Minusstunden eingetragen. Ansonsten wird unter 
Verwendung der Hilfsspalten für Pausen (R) und Tagesarbeitszeit (S) geprüft, ob die 
Tagesarbeitszeit unterschritten ist. In Zeile 37 wird dann die Gesamtsumme der 
Minusstunden gebildet.

Spalte O

=WENN(ODER(D6="";E6="");"";WENN((((WERT(E6)-WERT(D6))*24)-
R6)>10;"gesetzlich Tageshöchstarbeitszeit überschritten";""))

Falls die Tagesarbeitszeit (abzüglich Pausen) 10 Stunden überschreitet wird ein 
Warnhinweis ausgegeben.

Spalte R

=SUMME(WENN(F6="x";0,5;0);WENN(G6="x";1;0))

Hier werden, abhängig von den „x“ in den Spalten F und G die Pausenlängen ermittelt (½, 
1 oder 1,5 Stunden)



Spalte S

=WENN(UND($Basisdaten.$B$14="x";ODER(WOCHENTAG(A6)=1;WOCHE
NTAG(A6)=7));0;WENN(ODER(C6="";C6="ÜF";$Basisdaten.$B$14="x");$Ba
sisdaten.$B$11;WENN(ODER(C6="A";C6="T1";C6="F";C6="ÜFA";C6="ÜFT
1";C6="ÜFF");T6;WENN(ODER(C6="T2";C6="S";C6="ÜFT2";C6="ÜFS");U6;
WENN(ODER(C6="N";C6="ÜFN");V6;$Basisdaten.$B$11)))))

Wenn kein Schichtkürzel (A, T1, T2, F, S, N) angegeben ist, wenn ein freier Tag (ÜF) 
eingetragen ist (oder bei Redakteuren) ist die Tagesarbeitszeit die rechnerische tarifliche 
Tagesarbeitszeit (Bei Redakteuren am Samstag und Sonstag rechnerische 
Tagesarbeitszeit = 0).

Sonst wird auf die entsprechenden Tagesarbeitszeiten in den Spalten T, U und V 
zurückgegriffen (mit Fallback auf die tarifliche Tagesarbeitszeit) Dies gilt auch, wenn dem 
Schichtkürzel ein „ÜF“ für einen ganzen freien Tag vorangestellt wird.

WICHTIG: Teilzeitkräfte brauchen unbedingt einen hinterlegten Arbeitsplan, da sonst die 
Zuschlagsberechnung nicht hunderprozentig funktioniert.

Spalte T

=WENN(TEXT(A6;"TTT")="Mo";$Basisdaten.$E$21;WENN(TEXT(A6;"TTT")
="Di";$Basisdaten.$E$22;WENN(TEXT(A6;"TTT")="Mi";$Basisdaten.$E$23;
WENN(TEXT(A6;"TTT")="Do";$Basisdaten.$E$24;WENN(TEXT(A6;"TTT")="
Fr";$Basisdaten.$E$25;WENN(TEXT(A6;"TTT")="Sa";$Basisdaten.$E$26;W
ENN(TEXT(A6;"TTT")="So";$Basisdaten.$E$27;$Basisdaten.$B$11)))))))

Abhängig von den in Spalte A angezeigten Wochentagen wird aus den Arbeitsplänen die 
jeweilige Tagesarbeitszeit der Arbeitszeitmodelle A,T1 und F übertragen (mit Fallback auf 
die tarifliche Tagesarbeitszeit.

Spalten U und V

wie Spalte T für die Arbeitszeiten T2, S sowie N

Spalte W

=WENN(TEXT(A14;"TTT")="Do";SUMME(S11:S14;H11:H14);WENN(TEXT(
A14;"TTT")="Fr";SUMME(S10:S14;H10:H14);WENN(TEXT(A14;"TTT")="Sa";
SUMME(S9:S14;H9:H14);WENN(TEXT(A14;"TTT")="So";SUMME(S8:S14;H
8:H14);""))))

Um für Teilzeitkräfte feststellen zu können, ob die tarifliche Wochenarbeitszeit 
überschritten wurde (= Grenze zwischen Mehr- und Überstunden), muss ab Donnerstag 
eine entsprechende Prüfung erfolgen. Dabei wird am Monatsanfang auch auf die Daten 
aus dem Vormonat zurückgegriffen (nicht im Januar)



Spalte AB

=WENN((SUMME($H$6:H6)-SUMME($N$6:N6)+$P$38)<=0;"-";"+")

In dieser Hilfsspalte wird festgestellt, ob der aktuelle Tagessaldo zwischen Mehr- und 
Minusstunden (unter Einbeziehung des Vormonatssaldos) negativ oder positiv ist.

Spalte AC

=(SUMME($H$6:H6)-SUMME($N$6:N6)+$P$38)

Hier wird der aktuelle Tagessaldo als numerischer Wert ausgegeben.

Spalte AD

=SUMME(H6:H12)

In dieser Hilfsspalte wird festgestellt, wie groß die Summe der in den vergangenen 7 
Tagen geleisteten Mehrstunden ist.

Spalte AE

=WENN(AB6="+";"+";WENN(UND(AB6="-";AD6>10);"+";"-"))

Wenn der Tagessaldo (AB) positiv ist +, wenn der Tagessaldo negativ ist, aber mehr als 
10 Mehrstunden in den vergangenen 7 Tagen geleistet wurden (AD), dann +, sonst -.

Spalte AJ

=WENN(UND(AB7="+";AB6="+");H7;WENN(UND(AB7="+";AB6="-");H7+AC
6;WENN(UND(AB7="-";AB6="-";AD7>10);H7;"")))

Hier wird der zuschlagpflichtige Anteil der Mehrstunden errechnet. Dazu wird zunächst 
einmal geprüft, ob der aktuelle Tagessaldo und der Tagessaldo des Vortages positiv sind. 
In diesem Fall wwerden die Mehrstunden komplett zuschlagpflichtig. Beim Wechsel vom 
negativen in den positiven Tagessaldo wird von der Mehrstunden der negative 
Vortagsanteil abgezogen. Falls Tages- und Vortagessaldo negativ sind, wird geprüft, ob 
mehr als 10 Mehrstunden geleistet wurden (dann Zuschlagspflicht). Sonst bleibt die 
Spalte leer.


